

In[1]:= << HolonomicFunctions.m

HolonomicFunctions package by Christoph Koutschan, RISC-Linz, Version 1.3 (25.01.2010)
→ Type ?HolonomicFunctions for help

In[2]:= << Hyper.m

Zeilberger's holonomic ansatz reduces a determinant evaluation in question to proving three q-holonomic identities:

- (1) $c(n, n) = 1 \quad (n \geq 1)$
- (2) $\sum_{j=1}^n a(i, j) c(n, j) = 0 \quad (1 \leq i < n)$
- (3) $\sum_{j=1}^n a(n, j) c(n, j) = \frac{b(n)}{b(n-1)} \quad (n \geq 1)$

where $a(i, j)$ are the matrix entries ($1 \leq i, j \leq n$), $b(n)$ is the conjectured determinant evaluation (in the following we will refer to the quotient $b(n)/b(n-1)$ by "nice"), and the function $c(n, j)$ has to be pulled out of the hat.

Conjecture 1

The matrix entries:

```
a[i_Integer, j_Integer] :=  
  Sum[Binomial[2 (i - 1), k] * Binomial[2 (j - 1), k] * 4^k, {k, 0, 2 * (Min[i, j] - 1)}]  
  
aij = First[CreativeTelescoping[  
  Binomial[2 (i - 1), k] * Binomial[2 (j - 1), k] * 4^k, S[k] - 1, {S[i], S[j]}]]  
{ (5 i - 18 i^2 + 16 i^3 + 4 i j - 8 i^2 j) S_i + (-5 j - 4 i j + 18 j^2 + 8 i j^2 - 16 j^3) S_j +  
  (27 i + 18 i^2 - 144 i^3 - 27 j + 360 i^2 j - 18 j^2 - 360 i j^2 + 144 j^3),  
  (-5 + 8 i - 19 j + 24 i j - 22 j^2 + 16 i j^2 - 8 j^3) S_j^2 +  
  (165 - 584 i + 672 i^2 - 256 i^3 + 322 j - 720 i j + 384 i^2 j + 276 j^2 - 288 i j^2 + 80 j^3) S_j +  
  (81 j - 72 i j - 126 j^2 + 144 i j^2 - 72 j^3) }  
  
anj = DFiniteSubstitute[aij, {i -> n}, Algebra -> OreAlgebra[S[n], S[j]]];
```

Pull out of the hat an annihilating ideal for $c(n, j)$:

```
cnj = ToOrePolynomial[  
  { - ((-3 + 2 * j + 2 * n) * (-12 * j + 4 * j^2 + 72 * j^3 - 64 * j^4 + 18 * n + 27 * j * n - 172 * j^2 * n -  
    96 * j^3 * n + 256 * j^4 * n - 36 * n^2 + 198 * j * n^2 + 408 * j^2 * n^2 - 768 * j^3 * n^2 -  
    162 * n^3 - 432 * j * n^3 + 864 * j^2 * n^3 + 324 * n^4 - 432 * j * n^4)) + j * (-1 + 2 * j) *  
    (-1 + 4 * n) * (-60 + 196 * j - 200 * j^2 + 64 * j^3 - 27 * n + 36 * j * n + 54 * n^2 - 72 * j * n^2) *  
    S[j] + (-3 + 4 * j) * n * (1 - j + n) * (-1 + 2 * n) * (-3 + 4 * n) *  
    (-1 + 4 * n) * S[n], (1 + 4 * j) * (j - n) * (-3 + 2 * j + 2 * n) +  
    (11 - 8 * j + 4 * j^2 - 16 * j^3 - 27 * n + 36 * j * n + 18 * n^2 - 24 * j * n^2) * S[j] +  
    (1 + j) * (1 + 2 * j) * (-3 + 4 * j) * S[j]^2, OreAlgebra[S[n], S[j]] };
```

This function computes the initial values for $c(n, j)$:

```
mydet[n_Integer] := mydet[n] = Det[Table[a[i, j], {i, n}, {j, n}]]  
myc[n_Integer, j_Integer] := myc[n, j] = Which[j > n, 0, j == n == 1, 1, True,  
  (-1)^(n + j) * Det[Delete[#, j] & /@ Table[a[ii, jj], {ii, n - 1}, {jj, n}]] / mydet[n - 1]
```

Test whether recurrences and definition of $c(n, j)$ agree:

```

test = ApplyOreOperator[cnj, f[n, j]];
Table[test, {n, 5}, {j, 5}] /. f[n_, j_] -> myc[n, j]

{{{0, 0}, {0, 0}, {0, 0}, {0, 0}, {0, 0}},
 {{0, 0}, {0, 0}, {0, 0}, {0, 0}, {0, 0}}, {{0, 0}, {0, 0}, {0, 0}, {0, 0}, {0, 0}},
 {{0, 0}, {0, 0}, {0, 0}, {0, 0}, {0, 0}}, {{0, 0}, {0, 0}, {0, 0}, {0, 0}, {0, 0}}}

```

These points we need to give as initial values:

```

AnnihilatorSingularities[cnj, {1, 1}]

{{{j -> 1, n -> 1}, True}, {{j -> 2, n -> 1}, True}, {{j -> 2, n -> 2}, True}}

Clear[c];
c[1, 1] = myc[1, 1];
c[1, 2] = myc[1, 2];
c[2, 2] = myc[2, 2];
c[nn_Integer, jj_Integer] := c[nn, jj] =
  Module[{f, t},
    t = ApplyOreOperator[If[jj > 2,
      (S[n]^nn * S[j]^(jj - 2)) ** cnj[[2]], (S[n]^(nn - 1) * S[j]^jj) ** cnj[[1]]], f[n, jj]];
    t = t /. {j -> 0, n -> 0};
    Return[(-t /. f[nn, jj] -> 0 /. f -> c) / Coefficient[t, f[nn, jj]]];

```

Test whether this definition of $c(n,j)$ produces indeed the values that it is supposed to produce:

```

Union[Flatten[Table[c[n1, j1] == myc[n1, j1], {n1, 5}, {j1, 5}]]]
{True}

```

■ Proof of (1)

```

diag = Factor[OrePolynomialSubstitute[FindRelation[cnj, Pattern -> {a_, a_}], {S[j] -> 1, j -> n}]]

{n (1 + n)^2 (-1 + 2 n) (1 + 2 n)^2 (-3 + 4 n) (-1 + 4 n) (1 + 4 n) (3 + 4 n) (-105 + 1090 n - 3092 n^2 + 2320 n^3)
 S_n^2 - n (1 + n) (-1 + 2 n) (1 + 2 n) (-3 + 4 n) (-1 + 4 n) (-19485 - 29271 n + 906178 n^2 +
 1355300 n^3 - 7584920 n^4 - 11349904 n^5 + 9114592 n^6 + 13604480 n^7) S_n +
 18 n (1 + n) (-1 + 2 n) (1 + 2 n) (-1 + 3 n) (1 + 3 n) (-3 + 4 n) (-1 + 4 n) (-5 + 6 n)
 (-1 + 6 n) (213 + 1866 n + 3868 n^2 + 2320 n^3)}

OreReduce[diag, ToOrePolynomial[{S[n] - 1}]]

{0}

```

Initial values (note that there are no singularities in the leading coefficient):

```

Table[c[n1, n1], {n1, 2}]

{1, 1}

```

Hence $c(n,n)=1$ for all n .

■ Proof of (2)

```

Timing[
  sum = First[FindCreativeTelescoping[DFiniteTimes @@ (OreGroebnerBasis /@ ToOrePolynomial[
    {Append[aij, S[n] - 1], Append[cnj, S[i] - 1], OreAlgebra[S[i], S[j], S[n]]}],
    S[j] - 1]];]
{312.12, Null}

```

```
Support[sum]
```

```
{Si, Sn, 1}, {Sn3, Sn2, Sn, 1}}
```

Check initial values and singularities:

```
AnnihilatorSingularities[sum, {1, 1}, Assumptions → i < n]
```

```
{{{i → 1, n → 2}, True}, {{i → 1, n → 3}, True}, {{i → 1, n → 4}, True}}
```

```
MySum[a[i, j1] * c[n, j1], {j1, n}] /. (First /@%) /. MySum → Sum
```

```
{0, 0, 0}
```

This proves identity (2).

■ Proof of (3)

```
Timing[rec = First[FindCreativeTelescoping[DFiniteTimes[anj, cnj], S[j] - 1]]];
```

```
{125.256, Null}
```

```
Factor[rec]
```

$$\left\{ \begin{aligned} &(-3 + 4n)(-1 + 4n)(1 + 4n)(3 + 4n)^2(5 + 4n)(43 - 262n - 388n^2 + 2320n^3)S_n^2 - \\ &2(-3 + 4n)(-1 + 4n)(30960 + 105645n - 1300218n^2 - \\ &4876300n^3 + 4550040n^4 + 32729840n^5 + 38501088n^6 + 13604480n^7)S_n + \\ &72n(-1 + 2n)(-1 + 3n)(1 + 3n)(-5 + 6n)(-1 + 6n)(1713 + 5922n + 6572n^2 + 2320n^3) \end{aligned} \right\}$$

```
nice = 2^(4 * (n - 1)) * (2 (n - 1))! * (6 (n - 1))! * (3 n - 2) / ((4 (n - 1))!)^2 / (4 n - 3)
```

$$\frac{2^{4(-1+n)}(-2+3n)(2(-1+n))!(6(-1+n))!}{(-3+4n)((4(-1+n))!)^2}$$

```
OreReduce[rec, Annihilator[nice, S[n]]]
```

```
{0}
```

Compare initial values (note that there are no singularities in the leading coefficient of rec):

```
Table[Sum[a[n, j] * c[n, j], {j, n}] == nice, {n, 2}]
```

```
{True, True}
```

■ FIND the determinant evaluation automatically

We find a solution to the recurrence derived in (3), and verify that it indeed gives the result.

```
Hyper[ApplyOreOperator[First[rec], f[n]], f[n]]
```

Warning: irreducible factors of degree > 1 in leading
coefficient;
some solutions may not be found

Warning: irreducible factors of degree > 1 in trailing
coefficient;
some solutions may not be found

$$\left\{ \frac{36 (-1 + 3n) (1 + 3n) (-5 + 6n) (-1 + 6n)}{(-3 + 4n) (-1 + 4n)^2 (1 + 4n)} \right\}$$

`nicel = Product[First[%] /. n -> i, {i, n - 1}]`

$$\frac{2^{1-4n} 3^{-\frac{9}{2}+6n} \Gamma\left[-\frac{5}{3} + 2n\right] \Gamma\left[-\frac{1}{3} + 2n\right]}{\Gamma\left[-\frac{3}{2} + 2n\right] \Gamma\left[-\frac{1}{2} + 2n\right]}$$

Mathematica does not simplify this test to 0, but obviously both expressions agree.

`FullSimplify[nicel - nice]`

$$\frac{2^{1-4n} 3^{-\frac{9}{2}+6n} \Gamma\left[-\frac{5}{3} + 2n\right] \Gamma\left[-\frac{1}{3} + 2n\right]}{\Gamma\left[-\frac{3}{2} + 2n\right] \Gamma\left[-\frac{1}{2} + 2n\right]}$$

`Table[nicel - nice, {n, 20}] // FullSimplify`

`{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}`

Conjecture 2

Similar to Conjecture 1, we display only the most relevant computations.

```
cnj = ToOrePolynomial[
  {(-5 + 3 * j + 3 * n) * (48 * j + 150 * j^2 - 117 * j^3 - 810 * j^4 + 729 * j^5 - 96 * n - 328 * j * n +
    858 * j^2 * n + 2214 * j^3 * n - 2916 * j^4 * n + 144 * n^2 - 1698 * j * n^2 - 2880 * j^2 * n^2 +
    5184 * j^3 * n^2 + 1536 * n^3 + 2880 * j * n^3 - 5184 * j^2 * n^3 - 2304 * n^4 + 2592 * j * n^4) -
  j * (-2 + 3 * j) * (660 - 2811 * j + 4419 * j^2 - 2997 * j^3 + 729 * j^4 + 128 * n -
    528 * j * n + 432 * j^2 * n - 192 * n^2 + 792 * j * n^2 - 648 * j^2 * n^2) * S[j] +
  (-8 + 9 * j) * n * (1 - j + n) * (-2 + 3 * n) * (-1 + 3 * n) * (-5 + 6 * n) * S[n],
  (1 + 9 * j) * (j - n) * (-5 + 3 * j + 3 * n) -
  2 * (-67 + 67 * j - 15 * j^2 + 27 * j^3 + 160 * n - 180 * j * n - 96 * n^2 + 108 * j * n^2) * S[j] +
  (1 + j) * (1 + 3 * j) * (-8 + 9 * j) * S[j]^2}, OreAlgebra[S[n], S[j]]];
```

```

anj = First[CreativeTelescoping[
  Binomial[3 (n - 1), k] * Binomial[3 (j - 1), k] * 3^k, S[k] - 1, {S[n], S[j]}]]
{ (88 n + 54 j n + 54 j^2 n - 630 n^2 - 351 j n^2 - 243 j^2 n^2 + 1611 n^3 + 729 j n^3 +
  243 j^2 n^3 - 1782 n^4 - 486 j n^4 + 729 n^5) S_n + (-88 j + 630 j^2 - 1611 j^3 + 1782 j^4 -
  729 j^5 - 54 j n + 351 j^2 n - 729 j^3 n + 486 j^4 n - 54 j n^2 + 243 j^2 n^2 - 243 j^3 n^2) S_j +
  (-128 j - 144 j^2 + 1368 j^3 + 2592 j^4 - 5832 j^5 + 128 n - 4536 j^2 n - 3888 j^3 n + 21384 j^4 n + 144 n^2 +
  4536 j n^2 - 36936 j^3 n^2 - 1368 n^3 + 3888 j n^3 + 36936 j^2 n^3 - 2592 n^4 - 21384 j n^4 + 5832 n^5) ,
  (-88 - 538 j - 1143 j^2 - 1179 j^3 - 729 j^4 - 243 j^5 + 234 n + 1395 j n + 2700 j^2 n +
  2025 j^3 n + 486 j^4 n - 162 n^2 - 891 j n^2 - 1458 j^2 n^2 - 729 j^3 n^2) S_j^2 +
  (-11792 - 22900 j - 30276 j^2 - 22347 j^3 - 9720 j^4 - 1701 j^5 + 64620 n + 102672 j n +
  101763 j^2 n + 47952 j^3 n + 9963 j^4 n - 142236 n^2 - 169533 j n^2 - 110808 j^2 n^2 -
  24786 j^3 n^2 + 157221 n^3 + 122472 j n^3 + 39366 j^2 n^3 - 87480 n^4 - 32805 j n^4 + 19683 n^5) S_j +
  (1568 j - 5760 j^2 + 1656 j^3 + 3888 j^4 + 1944 j^5 - 2736 j n + 11448 j^2 n -
  8424 j^3 n - 3888 j^4 n + 1296 j n^2 - 5832 j^2 n^2 + 5832 j^3 n^2) }

Timing[rec = First[FindCreativeTelescoping[DFiniteTimes[anj, cnj], S[j] - 1]];]
{326.56, Null}

Factor[rec]
{-2 (-1 + 3 n) (2 + 3 n)^2 (-5 + 6 n) (1 + 6 n) (7 + 6 n) (442 - 1029 n - 7056 n^2 + 16605 n^3) S_n^2 +
  3 (-1 + 3 n) (-5 + 6 n) (1113840 + 15040708 n - 2282346 n^2 - 306529938 n^3 -
  368213553 n^4 + 1032343704 n^5 + 2056776543 n^6 + 917293410 n^7) S_n +
  108 n (-2 + 3 n) (-1 + 4 n) (1 + 4 n) (-11 + 12 n) (-5 + 12 n) (8962 + 34674 n + 42759 n^2 + 16605 n^3) }

nice = 2^(8 * (n - 1)) * Pochhammer[7 / 12, n - 1] * Pochhammer[1 / 12, n - 1] * Pochhammer[5 / 4, n - 1] *
  Pochhammer[3 / 4, n - 1] / Pochhammer[7 / 6, n - 1] / Pochhammer[1 / 6, n - 1] /
  Pochhammer[2 / 3, n - 1] / Pochhammer[2 / 3, n - 1]

2^8 (-1+n) Pochhammer[1/12, -1+n] Pochhammer[7/12, -1+n] Pochhammer[3/4, -1+n] Pochhammer[5/4, -1+n]
-----
Pochhammer[1/6, -1+n] Pochhammer[2/3, -1+n]^2 Pochhammer[7/6, -1+n]

OreReduce[rec, Annihilator[nice, S[n]]]
{0}

Hyper[ApplyOreOperator[First[rec], f[n]], f[n]]

```

Warning: irreducible factors of degree > 1 in leading
coefficient;
some solutions may not be found

Warning: irreducible factors of degree > 1 in trailing
coefficient;
some solutions may not be found

$$\left\{ \frac{36 (-1 + 4 n) (1 + 4 n) (-11 + 12 n) (-5 + 12 n)}{(-1 + 3 n)^2 (-5 + 6 n) (1 + 6 n)} \right\}$$

```

nicel = Product[First[%] /. n -> i, {i, n - 1}]


$$\frac{2^{-\frac{20}{3}+8n} \Gamma\left[-\frac{11}{6} + 2n\right] \Gamma\left[-\frac{1}{2} + 2n\right]}{3 \sqrt{3} \Gamma\left[-\frac{5}{3} + 2n\right] \Gamma\left[-\frac{2}{3} + 2n\right]}$$


FullSimplify[nicel - nice]

{0}

```

Bacher determinant

The matrix entries:

```

In[4]:= a[i_Integer, j_Integer] :=
Sum[Binomial[2 (i - 1), k] * Binomial[2 (j - 1), k] * 2^k, {k, 0, 2 * (Min[i, j] - 1)}]

In[6]:= aij = First[CreativeTelescoping[
Binomial[2 (i - 1), k] * Binomial[2 (j - 1), k] * 2^k, S[k] - 1, {S[i], S[j]}]]

Out[6]=  $\left\{ \left( 3 i - 10 i^2 + 8 i^3 \right) S_i + \left( -3 j + 10 j^2 - 8 j^3 \right) S_j + \left( i + 2 i^2 - 8 i^3 - j + 16 i^2 j - 2 j^2 - 16 i j^2 + 8 j^3 \right), \right.$ 
 $\left. \left( 1 + 3 j + 2 j^2 \right) S_j^2 + \left( -5 + 12 i - 8 i^2 - 2 j - 4 j^2 \right) S_j + \left( -j + 2 j^2 \right) \right\}$ 

In[7]:= anj = DFiniteSubstitute[aij, {i -> n}, Algebra -> OreAlgebra[S[n], S[j]]];

```

Pull out of the hat an annihilating ideal for $c(n,j)$:

```

In[8]:= cnj = ToOrePolynomial[
{(-1 + 2 * j) * (1 + 4 * j) * (j - n) - j * (-3 + 4 * j) * (-1 + 2 * j + 2 * n) * S[j], n * (-3 + 4 * n) *
(-1 + 4 * n) - (-1 + j - n) * (-1 + 2 * n) * (-1 + 2 * j + 2 * n) * S[n]}, OreAlgebra[S[n], S[j]]];

```

This function computes the initial values for $c(n,j)$:

```

In[9]:= mydet[n_Integer] := mydet[n] = Det[Table[a[i, j], {i, n}, {j, n}]]
myc[n_Integer, j_Integer] := myc[n, j] = Which[j > n, 0, j == n == 1, 1, True,
(-1)^(n + j) * Det[Delete[#, j] & /@ Table[a[ii, jj], {ii, n - 1}, {jj, n}]] / mydet[n - 1]]

```

Test whether recurrences and definition of $c(n,j)$ agree:

```

In[11]:= test = ApplyOreOperator[cnj, f[n, j]];
Table[test, {n, 5}, {j, 5}] /. f[n_, j_] -> myc[n, j]

Out[12]=  $\{\{ \{0, 0\}, \{0, 0\}, \{0, 0\}, \{0, 0\}, \{0, 0\} \},$ 
 $\{ \{0, 0\}, \{0, 0\}, \{0, 0\}, \{0, 0\}, \{0, 0\} \}, \{ \{0, 0\}, \{0, 0\}, \{0, 0\}, \{0, 0\}, \{0, 0\} \},$ 
 $\{ \{0, 0\}, \{0, 0\}, \{0, 0\}, \{0, 0\}, \{0, 0\} \}, \{ \{0, 0\}, \{0, 0\}, \{0, 0\}, \{0, 0\}, \{0, 0\} \} \}$ 

```

These points we need to give as initial values:

```

In[13]:= AnnihilatorSingularities[cnj, {1, 1}]

Out[13]=  $\{\{j \rightarrow 1, n \rightarrow 1\}, \text{True}\}$ 

In[22]:= LeadingPowerProduct /@ cnj

Out[22]=  $\{S_j, S_n\}$ 

```

```
In[23]:= Clear[c];
c[1, 1] = myc[1, 1];
c[nn_Integer, jj_Integer] := c[nn, jj] =
Module[{f, t},
  t = ApplyOreOperator[If[jj > 1,
    (S[n]^nn * S[j]^(jj - 1)) ** cnj[[1]], (S[n]^(nn - 1) * S[j]^jj) ** cnj[[2]]], f[n, jj]];
  t = t /. {j -> 0, n -> 0};
  Return[(-t /. f[n, jj] -> 0 /. f -> c) / Coefficient[t, f[nn, jj]]];
```

Test whether this definition of $c(n,j)$ produces indeed the values that it is supposed to produce:

```
In[27]:= Union[Flatten[Table[c[n1, j1] === myc[n1, j1], {n1, 5}, {j1, 5}]]]
Out[27]:= {True}
```

■ Proof of (1)

```
In[28]:= diag = Factor[OrePolynomialSubstitute[FindRelation[cnj, Pattern -> {a_, a_}], {S[j] -> 1, j -> n}]]
Out[28]:= {-n (-1 + 2 n) (-3 + 4 n) (-1 + 4 n) (1 + 4 n) S_n + n (-1 + 2 n) (-3 + 4 n) (-1 + 4 n) (1 + 4 n)}
In[29]:= OreReduce[diag, ToOrePolynomial[{S[n] - 1}]]
Out[29]:= {0}
```

Initial values (note that there are no singularities in the leading coefficient):

```
In[30]:= Table[c[n1, n1], {n1, 2}]
Out[30]:= {1, 1}
```

Hence $c(n,n)=1$ for all n .

■ Proof of (2)

```
In[31]:= Timing[
  sum = First[FindCreativeTelescoping[DFiniteTimes @@ (OreGroebnerBasis /@ ToOrePolynomial[
    {Append[aij, S[n] - 1], Append[cnj, S[i] - 1]}, OreAlgebra[S[i], S[j], S[n]]),
    S[j] - 1]];]
Out[31]:= {4.90831, Null}
```

```
In[32]:= Support[sum]
Out[32]:= {{S_i, S_n, 1}, {S_n^2, S_n, 1}}
```

Check initial values and singularities:

```
In[33]:= AnnihilatorSingularities[sum, {1, 1}, Assumptions -> i < n]
Out[33]:= {{i -> 1, n -> 2}, True}, {{i -> 1, n -> 3}, True}}
In[34]:= MySum[a[i, j1] * c[n, j1], {j1, n}] /. (First /@ %) /. MySum -> Sum
Out[34]:= {0, 0}
```

This proves identity (2).

■ Proof of (3)

```
In[35]:= Timing[rec = First[FindCreativeTelescoping[DFiniteTimes[anj, cnj], S[j] - 1]];]
```

```
Out[35]:= {1.76411, Null}
```

```
In[36]:= Factor[rec]
```

```
Out[36]:= {-(-1 + 2 n)^2 S_n + 4(-3 + 4 n)(-1 + 4 n)}
```

```
In[37]:= nice = 2^(2*(n-1)) * ((n-1)!)^2 * (4(n-1))! / ((2(n-1))!)^3
```

```
Out[37]:= 
$$\frac{2^{2(-1+n)} ((-1+n)!)^2 (4(-1+n))!}{((2(-1+n))!)^3}$$

```

```
In[38]:= OreReduce[rec, Annihilator[nice, S[n]]]
```

```
Out[38]:= {0}
```

Compare initial values (note that there are no singularities in the leading coefficient of rec):

```
In[39]:= Table[Sum[a[n, j] * c[n, j], {j, n}] == nice, {n, 2}]
```

```
Out[39]:= {True, True}
```

■ FIND the determinant evaluation automatically

We find a solution to the recurrence derived in (3), and verify that it indeed gives the result.

```
In[40]:= Hyper[ApplyOreOperator[First[rec], f[n]], f[n]]
```

```
Out[40]:= 
$$\left\{ \frac{4(-3 + 4n)(-1 + 4n)}{(-1 + 2n)^2} \right\}$$

```

```
In[41]:= nice1 = Product[First[%] /. n -> i, {i, n - 1}]
```

```
Out[41]:= 
$$\frac{4^{-1+n} \sqrt{\pi} \text{Gamma}\left[-\frac{3}{2} + 2n\right]}{\text{Gamma}\left[-\frac{1}{2} + n\right]^2}$$

```

```
In[42]:= FullSimplify[nice1 - nice]
```

```
Out[42]:= 0
```